

A Path to Gain Confidence with your Confidence Interval in Clinical Trial Analysis

Yi Gu, Roche (China) Holding Ltd

ABSTRACT

In the statistical analysis of clinical trials, a confidence interval provides useful information regarding to the precision of both the efficacy and safety outcomes. Different method of confidence interval calculation will lead to deviations of the results. A better understanding of the confidence interval calculation methods in SAS functions is very important to SAS programmers in generating the right numbers based on the different algorithms suggested by statisticians. This paper is to present the SAS functions which are frequently implemented in computing CI for means or proportion under different scenarios of clinical trials.

INTRODUCTION

A confidence interval is an estimate range of values for the parameter of interest. In clinical trials, it provides the information to interpret the means, proportions and model parameters. From a statistical programmer's point of view, understanding the default settings and the potential adjustments in the SAS functions is helpful to provide efficient and accurate computation of confidence intervals based on the statistical analysis plan.

In this article, we will discuss the SAS function commonly used under different scenario and give the audience a general idea of the implementation.

CONFIDENCE INTERVAL FOR NUMERIC VARIABLES

In the descriptive summary table of subject demographics in clinical trials, confidence interval for means is an important statistics for the data interpretation. The confidence limit presents a range, constructed around the value of a sample mean that contains the corresponding true population mean with given probability in repeated sampling.

PROC MEANS <OPTION(S)> <CLM| UCLM| LCLM>;

PROC MEANS procedure produces simple univariate descriptive statistics for numeric variables.

Statistic keywords for confidence interval computation include UCLM, LCLM, and CLM. CLM keyword can request two-sided confidence limits, and UCLM and LCLM request one-side confidence limits for the Student's t distribution:

```
proc means data=ars_test n mean clm;  
    var bage;  
run;
```

Analysis Variable : BAGE Baseline Age					
trt01p	N Obs	N	Mean	Lower 95% CL for Mean	Upper 95% CL for Mean
arm A	103	103	61.1262136	58.9246260	63.3278012
arm B	95	95	65.9052632	63.7896874	68.0208390

Output 1. Output from a PROC MEANS Statement

PROC UNIVARIATE

<OPTION(S)> <CIBASIC|CIPCTLDF|CIQUANTDF|CIPCTLNORMAL|CIQUANTNORMAL>;

PROC UNIVARIATE can be used to request a variety of statistics for summarizing the data distribution of confidence intervals for the mean and quantiles.


```
proc univariate data=ars_test (where=(trt01p="arm A")) cibasic;
  var bage;
run;
```

Basic Confidence Limits Assuming Normality			
Parameter	Estimate	95% Confidence Limits	
Mean	61.12621	58.92463	63.32780
Std Deviation	11.26480	9.90841	13.05485
Variance	126.89568	98.17659	170.42913

Output 2. Output from a PROC UNIVARIATE Statement

The CIBASIC option requests confidence limits under the assumption that the data are normally distributed. Before using the confidence interval, the assumption should be checked by the Shapiro-Wilk test for normality in PROC UNIVARIATE.

```
proc univariate data=ars test(where=(trt01p="arm A"));
  histogram bage / normal(percents=20 40 60 80 midpercents);
  inset n normal(ksdpval) / pos = ne format = 6.3;
run;
```



Output 3. Output Testing Normal Distribution

In output 3, $p > 0.01$ indicates the normal distribution of the data, which means we can use CIBASIC to compute the corresponding confidence interval. In other cases, if the assumption of abnormality is rejected, the CIQUANTDF is a better option for the computation of distribution-free confidence limits for the quantiles of baseline ages. Here, we still use the same data set for comparison.

```
proc univariate data=ars_test (where=(trt01p="arm A")) ciquantdf;
  var bage;
run;
```

Quantiles (Definition 5)						
Level	Quantile	95% Confidence Limits Distribution Free		Order Statistics		
		LCL Rank	UCL Rank	Coverage		
100% Max	80					
99%	80	80	80	101	103	55.99
95%	79	77	80	94	103	96.13
90%	74	73	79	87	99	95.40
75% Q3	72	68	73	68	86	95.81
50% Median	62	58	64	42	62	95.18
25% Q1	53	49	57	18	36	95.81
10%	45	43	46	5	17	95.40
5%	43	39	45	1	10	96.13
1%	39	39	39	1	3	55.99
0% Min	39					

Output 4. Output from PROC UNIVARIATE<CIQUANTDF> statement.

CONFIDENCE INTERVAL FOR CATEGORICAL VARIABLES

In clinical trials, binomial proportion and its confidence interval estimation provides important information for the analysis of categorical endpoints.

PROC FREQ <OPTION(S)>;

BY VARIABLES;

EXACT BINOMIAL </ COMPUTATION-OPTIONS>;

OUTPUT < OUT=SAS-DATA-SET > OUTPUT-OPTIONS;

TABLES REQUESTS </ BINOMIAL (CL =)>;

WEIGHT VARIABLES;

PROC FREQ is a commonly used procedure to compute the binomial proportion for one-way tables.

```
proc freq data=ars_test;
  table aval/binomial;
  by trt01p;
run;
```

The FREQ Procedure

trt01p=arm A

response	Frequency	Percent	Cumulative Frequency	Cumulative Percent
0	54	52.43	54	52.43
1	49	47.57	103	100.00

Binomial Proportion	
response = 0	
Proportion	0.5243
ASE	0.0492
95% Lower Conf Limit	0.4278
95% Upper Conf Limit	0.6207
Exact Conf Limits	
95% Lower Conf Limit	0.4235
95% Upper Conf Limit	0.6236

Output 5. Output from a PROC FREQ Statement.

PROC FREQ computes the confidence interval under the following default settings. These setting usually can be adjusted depending on multiple requirements of statistical analysis. On the hand, ignorance of the default settings may cause some practical difficulties in implementation.

Default 1: ALPHA = 0.05

```
proc freq data=ars_test;
  table aval/binomial alpha=0.1;
  by trt01p;
run;
```

The ALPHA= option enables the adjustment of the level of the confidence limits.

Default 2: provides Wald and exact (Clopper-Pearson) confidence limits for the binomial proportion

Phase II clinical studies are designed to determine whether there is sufficient evidence of efficacy and safety to make it worth for further study with a larger and expensive scale. They are often based on a relatively small population which requests a proper type of analysis for confidence interval based on the sample size and distribution.

By specifying CL=ALL, six types of confidence limits requested, including Agresti-Coull, Exact (Clopper-Pearson), Jeffreys, Wald and Wilson. Other types can also be specified.

```
proc freq data=ars_test;
  table response/binomial (cl=all);
  by trt01p;
run;
```

Confidence Limits for the Binomial Proportion		
Proportion = 0.5243		
Type	95% Confidence Limits	
Agresti-Coull	0.4287	0.6181
Clopper-Pearson (Exact)	0.4235	0.6236
Jeffreys	0.4283	0.6189
Wald	0.4278	0.6207
Wilson	0.4287	0.6181

Output 5. Output from PROC FREQ statement listing CL=ALL

```
proc freq data=ars_test;
  table response/binomial(cl= WALD
    WALD (CORRECT)
    WILSON
    WILSON (CORRECT)
    CLOPPERPEARSON
    MIDP
    LIKELIHOODRATIO
    JEFFREYS
    AGRESTICOULL
    LOGIT
    BLAKER) ;
  by trt01p;
run;
```

Confidence Limits for the Binomial Proportion		
Proportion = 0.5579		
Type	95% Confidence Limits	
Agresti-Coull	0.4577	0.6536
Blaker	0.4550	0.6595
Clopper-Pearson (Exact)	0.4523	0.6598
Jeffreys	0.4575	0.6548
Likelihood Ratio	0.4575	0.6552
Logit	0.4570	0.6542
Mid-p	0.4570	0.6553
Wald (Corrected)	0.4528	0.6630
Wilson (Corrected)	0.4525	0.6585

Output 6. Output from PROC FREQ statement listing specified methods for CI computation

Default 3: the result appears in the output is the proportion of observation in the first variable level.

As is presented in the statistical table in output 5, the default setting of PROC FREQ computes the confidence interval for RESPONSE=0 instead of RESPONSE=1 of interest. A solution frequently implemented to this, is inverting the two categorical levels to make RESPONSE=1 as the lowest level that PROC FREQ can compute the confidence interval.

```
data ars_test;
  set ars test;
  if response=0 then response=2;
  proc sort; by trt01p;
```

```
run;

proc freq data=ars_test;
  table response/binomial;
  by trt01p;
run;
```

trt01p=arm A				
response	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	49	47.57	49	47.57
2	54	52.43	103	100.00

Binomial Proportion	
response = 1	
Proportion	0.4757
ASE	0.0492
95% Lower Conf Limit	0.3793
95% Upper Conf Limit	0.5722
Exact Conf Limits	
95% Lower Conf Limit	0.3764
95% Upper Conf Limit	0.5765

Output 7. Output from PROC FREQ statement with adjusted response level

After the adjustment, the lower and upper 95% confidence limits in the output 7 show that the response rate of the treatment lies between (0.38, 0.68).

Default 4: do not present the level with zero proportion.

trt01p=arm B				
response	Frequency	Percent	Cumulative Frequency	Cumulative Percent
0	95	100.00	95	100.00

Binomial Proportion	
response = 0	
Proportion	1.0000
ASE	0.0000
95% Lower Conf Limit	1.0000
95% Upper Conf Limit	1.0000
Exact Conf Limits	
95% Lower Conf Limit	0.9619
95% Upper Conf Limit	1.0000

Output 8. Output from PROC FREQ statement without presenting the level with zero

Under this situation, a commonly suggested way is to call the option of ZERO in WEIGHT. The first step is to derive a summary table for the count based on current data. Then the creation of the *freqcat* dataset is to set back the missing row of zero count.

```

data freqcat;
  input trt01p $1-5 response 7;
  cards;
arm A 1
arm A 2
arm B 1
arm B 2
  ;
proc sort; by trt01p response;
run;

proc sort data=freqtable;
  by trt01p response;
run;

data freq;
  merge freqtable freqcat;
  by trt01p response;
  if count=. then count=0;
  proc sort; by trt01p;
run;

proc freq data=freq;
  table response/binomial;
  by trt01p;
  weight count/zero;
run;

```

trt01p=arm B				
response	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	0	0.00	0	0.00
2	95	100.00	95	100.00

Binomial Proportion	
response = 1	
Proportion	0.0000
ASE	0.0000
95% Lower Conf Limit	0.0000
95% Upper Conf Limit	0.0000
Exact Conf Limits	
95% Lower Conf Limit	0.0000
95% Upper Conf Limit	0.0381

Test of H0: Proportion = 0.5	
ASE under H0	0.0513
Z	-9.7468
One-sided Pr < Z	<.0001
Two-sided Pr > Z	<.0001

Output 8. Output from PROC FREQ statement with presenting the level with zero after adjustment

As is presented, treatment group B, which did not present in the statistical table due to the zero response rate is listed in the table after the adjustment, with the exact confidence interval of (0, 0.038) .

```
PROC SURVEYFREQ <OPTION(S)>;
BY VARIABLES;
CLUSTER VARIABLES;
REPWEIGHTS VARIABLES < /OPTIONS>;
STRATA VARIABLES < /OPTIONS>;
TABLES REQUESTS </ OPTIONS>;
WEIGHT VARIABLES;
```

Apart from PROC FREQ, PROC SURVEYFREQ procedure produces one-way to n-way frequency and crosstabulation tables including confidence limits. The difference is that PROC SURVEY enables sample design information including different sampling rates or totals for different strata. It provides more accurate statistics with the consideration of the strata and clusters.

Input data type	Tests and parameters for computation with design-adjusted
One-way frequency tables	Rao-Scott chi-square goodness-of-fit
Two-way frequency tables	Rao-Scott likelihood ratio test Wald chi-square Wald log-linear chi-square
2 X 2 tables	Risks Risk difference Odds ratio Relative risks

Table 1. Data types computable in PROC SURVEYFREQ statement

CONFIDENCE INTERVAL FOR MODEL PARAMETERS

Fitting a generalized linear model, PROC GENMOD produces the estimation of the parameters of the model and the corresponding confidence interval of them. In this part, we use the example in SAS/STAT(R) 9.2 User's Guide to show how PROC GENMOD generates the confidence interval in a logistic regression. The distribution involved here is a binomial distribution. The case discussed the effects of multi-drugs clinical trial on a number of subjects. The outcome is defined as a binomial response. X represents the covariates in the model. The data set is built as the following:

```
data drug;
  input drug$ x response n @@;
  datalines;
  A .1 1 10 A .23 2 12 A .67 1 9
  B .2 3 13 B .3 4 15 B .45 5 16 B .78 5 13
  C .04 0 10 C .15 0 11 C .56 1 12 C .7 2 12
  D .34 5 10 D .6 5 9 D .7 8 10
  E .2 12 20 E .34 15 20 E .56 13 15 E .8 17 20
  ;
run;
```



```
proc genmod data=drug;
  class drug;
  model response/n = x drug / dist = bin link = logit lrci;
run;
```

Analysis Of Maximum Likelihood Parameter Estimates								
Parameter		DF	Estimate	Standard Error	Likelihood Ratio 95% Confidence Limits		Wald Chi-Square	Pr > ChiSq
Intercept		1	0.2792	0.4196	-0.5336	1.1190	0.44	0.5057
x		1	1.9794	0.7660	0.5038	3.5206	6.68	0.0098
drug	A	1	-2.8955	0.6092	-4.2280	-1.7909	22.59	<.0001
drug	B	1	-2.0162	0.4052	-2.8375	-1.2435	24.76	<.0001
drug	C	1	-3.7952	0.6655	-5.3111	-2.6261	32.53	<.0001
drug	D	1	-0.8548	0.4838	-1.8072	0.1028	3.12	0.0773
drug	E	0	0.0000	0.0000	0.0000	0.0000	.	.
Scale		0	1.0000	0.0000	1.0000	1.0000		

Output9. Output from PROC GENMOD statement

In output 9, PROC GENMOD produced the 95% confidence interval for likelihood ratio of each parameter in the logistic regression model. The computation here is enabled by using the LRCI option.

CONCLUSION

This paper presented a very basic summary of the tips when computing confidence intervals with commonly used SAS functions, which enables different methods and draw distinct results. It is very important for the greener to be aware that the default settings cannot meet all the requirements defined in the statistical analysis plan in clinical trial. Even the very basic function, such as PROC MEANS and PROC FREQ need careful handling in terms of different data distribution. The adjustment of dataset structure and SAS options is helpful for derivation of the accurate confidence interval.

REFERENCES

- Attain 100% Confidence in Your 95% Confidence Interval, Indu Nair, Binal Patel, PharmaSUG 2014 - Paper IB05
- Confidence Intervals Are a Programmer's Friend, Xinxin Guo, Zhaohui Su, PharmaSUG 2015 – Paper SP06
- Computation of CIs for Binomial proportions in SAS and its practical difficulties, Jose Abraham, Phuse 2013 – Paper SP05

RECOMMENDED READING

- *Base SAS® Procedures Guide*

CONTACT INFORMATION <HEADING 1>

Your comments and questions are valued and encouraged. Contact the author at:

Name: Yi Gu
Enterprise: Roche (China) Holding Ltd
Email: ivy.gu@roche.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.